Pensieve Header: Plotting 2-variable polynomials.

```
Print[
 "Loading PolyPlot.m from http://drorbn.net/ktc25/ap to plot 2-variable polynomials."]
```

```
Loading PolyPlot.m from http://drorbn.net/ubc24/ap to plot 2-variable polynomials.
```

```
Options[PolyPlot1] = {Labeled → False};
PolyPlot1[Δ_, OptionsPattern[]] := Module[{crs, m, maxc, minc, s, rect},
  rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
  If[Expand[Δ] === 0, Graphics[],
   m = Max[-Exponent[Δ, T, Min], Exponent[Δ, T, Max]];
   crs = CoefficientRules[T^m Δ, {T}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[crs /. ({x_} → c_) :> {
       Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
       Tooltip[Polygon[({x + m - 1 / 2, 0} + #) & /@ rect], c T^(x-m)],
       If[Not@OptionValue[Labeled], {},
        {Black, FontSize → Scaled[1/((m + 1) (6 + maxc))], Text[c T^(x-m), {x + m, 0.5}]}]
      }, AspectRatio → Min[1 / 5, 1 / Sqrt[m + 1]],
    ImagePadding → None, PlotRangePadding → None]
  ]];
Options[PolyPlot2] = {Labeled → False};
PolyPlot2[Θ_, OptionsPattern[]] := Module[{crs, m1, m2, maxc, minc, s, hex, p},
   If[Expand[Θ] === 0, Graphics[{White, Disk[]}],
    hex = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2, {α, 2 π / 12, 2 π, 2 π / 6}];
    m1 = Max[-Exponent[Θ, T1, Min], Exponent[Θ, T1, Max]];
    m2 = Max[-Exponent[Θ, T2, Min], Exponent[Θ, T2, Max]];
    crs = CoefficientRules[T1^m1 T2^m2 Θ, {T1, T2}];
    maxc = N@Log@Max@Abs[Last /@ crs];
    minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
    If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
    Graphics[{(*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/Sqrt[2]]},*)
      crs /. ({x1_, x2_} → c_) :> {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        p = {{1, -1/2}, {0, Sqrt[3]/2}}.{x1 - m1, x2 - m2};
        Tooltip[Polygon[(p + #) & /@ hex], c T1^(x1-m1) T2^(x2-m2)],
```

```
          If[Not@OptionValue[Labeled], {},
           {Black, FontSize → Scaled[1/(Max[m1, m2] (10 + maxc))], Text[c T1^(x1-m1) T2^(x2-m2), p]}]
         }
       }, ImagePadding → None, PlotRangePadding → None]
     ]];
PolyPlot[{Δ_, Θ_}, opts___Rule] := GraphicsColumn[
    {PolyPlot1[Δ, FilterRules[{opts}, Options[PolyPlot1]]],
     PolyPlot2[Θ, FilterRules[{opts}, Options[PolyPlot2]]]},
    Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
    FilterRules[{opts}, Options[GraphicsColumn]]
   ];
```

In[•]:=
$$\text{PolyPlot@}\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2\,T_2^2}+\frac{1}{T_1\,T_2^2}+\frac{1}{T_1^2\,T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2\,T_2-T_2^2+T_1\,T_2^2-T_1^2\,T_2^2\right\}$$

Out[•]=